
分岐・ジャンプ命令

条件文や繰り返し文などを実現するには，命令の実行順の制御
が必要

(C言語)

(MIPSアセンブリ言語)

if (x == y) {

x = x + 1;

}

...

bne   $s0, $s1, L1    # $s0 ≠ $s1 ならば L1 へ分岐
addu  $s0, $s0, 1

L1: ...

ただし，変数 x, y の内容がそれぞれレジスタ
s0, s1 に置かれているとする

ラベル

資料: 主な分岐命令，比較命令

命令 説明

beq $a, $b, L $a = $b ならばラベル L へ分岐 (branch on equal)

bne $a, $b, L $a ≠$b 〃 (branch on not equal)

slt $c, $a, $b 符号つきで $a < $b ならば $c ← 1;  さもなくば $c ← 0 

sltu $c, $a, $b 符号無しで $a < $b ならば $c ← 1;  さもなくば $c ← 0

sgt $c, $a, $b 符号つきで $a > $b 〃 (set on greater than)

sgtu $c, $a, $b 符号無しで $a > $b 〃 (set on greater than unsigned)

sle $c, $a, $b 符号つきで $a≦$b 〃 (set on less than or equal)

sleu $c, $a, $b 符号無しで $a≦$b 〃 (set on less than or equal unsigned)

sge $c, $a, $b 符号つきで $a≧$b 〃 (set on greater than or equal)

sgeu $c, $a, $b 符号無しで $a≧$b 〃 (set on greater than or equal  unsigned)

•slt, sltu 以外の比較命令はマクロ命令

例: if – else 文

if (x < y) {

x = x + 1;

} else {

x = x + 2;

}

...

slt   $t0, $s0, $s1

beq   $t0, $zero, L1  # $s0 < $s1 でないならL1へ分岐
addu  $s0, $s0, 1

j     L2              # 無条件に L2 へジャンプ
L1: addu  $s0, $s0, 2

L2: ...

(C言語)

(MIPSアセンブリ言語)

ただし，変数 x, y の内容がそれぞれレジスタ
s0, s1 に置かれているとする

例: while文

while (x < y) {

x = x + 1;

}

L1: slt $t0, $s0, $s1    # x < y なら $t0 ← 1; さもなくば $t0 ← 0
beq $t0, $zero, L2 # 比較結果が偽（ゼロ）なら L2 へ
addu $s0, $s0, 1 # x ← x + 1
j L1

L2: ...

(C言語)

(MIPSアセンブリ言語)

ただし，変数 x, y の内容がそれぞれレジスタ
s0, s1 に置かれているとする

関数呼出し

int func(int a, int b) {

a = a + b;

return a;

}

int main() {

int x, y, z;

x = func(5, 1);

y = func(8, 2);

z = func(x, y);

...

}

(C言語)

ただし，関数main内の変数 x, y, z の内
容がそれぞれレジスタ s0, s1, s2 に置か
れているとする

関数呼出しは単に j 命令でジャンプ
するだけでは実現できない
（元の位置に戻って来る必要がある）

関数呼出しの実行例

func: addu $a0, $a0, $a1

move $v0, $a0       # 戻り値には $v0, $v1 を使う
jr   $ra            # $ra のアドレスへジャンプ

main:                     # 初期化省略
li   $a0, 5         # 引数には $a0～$a3 を使う
li   $a1, 1

jal func     # func へジャンプし，戻り先アドレスを $ra へ
move $s0, $v0      

li $a0, 8    

li $a1, 2      

jal func

move $s1, $v0

move $a0, $s0 

move $a1, $s1 

jal func

move $s2, $v0

(MIPSアセンブリ言語)



8鏡 慎吾 (東北大学): 情報科学基礎I 2018 (7)

資料: 主なジャンプ命令

命令 説明

j      L ラベル L へジャンプ (jump)

jal   L ラベル L へジャンプすると同時に，次の命令アドレ
スを $ra ($31) に保存
(jump and link)

jr    $r レジスタ r に保存されたアドレスへジャンプ
(jump register)

分岐・ジャンプ命令の動作

メモリ

32ビットALU 

32x32ビット
レジスタ

PC

命令デコーダ

アドレス（32ビット）

データ（8, 16, 32ビット）

次PC計算

制御回路

mux

mux

(1) レジスタ
読み出し

(2) 分岐判定

(3) 分岐先アドレス計算

参考: 「次PC計算」の処理

分岐条件が成立していないなら，
• PC + 4

成立しているなら，
• 条件分岐命令のとき:

PC + 4 £ 定数フィールド値
• ジャンプ命令のとき: 

PC の下位28ビットを 4 £ 定数フィールド値 で置換

※命令長は常に4バイトなので，分岐・ジャンプ先のアドレスは必ず
4の倍数になる

※ この講義では「遅延分岐」は無視する

MIPS の命令フォーマット

MIPSの命令は基本的に以下の3フォーマットに分類される

31            26 25        21 20       16 15         11 10         6  5               0

funcOP rs rt rd シフト量

31            26 25        21 20       16 15                                                 0

OP rs rt 定数（即値 / アドレスオフセット）

31           26 25                                                                                 0

OP 定数（ジャンプ先）

R形式

I形式

J形式

レジスタ間演算，シフト命令（即値版も含む）など

レジスタ-即値間演算，ロード・ストア，分岐命令など

ジャンプ命令など

実際の関数呼出し

前の関数呼出しの例は，簡単に済むように巧妙に作られた例
である．実際には以下のようなことを考えなくてはならない．

• 呼出された関数はどのレジスタを使えばよいのか? 特に，呼
出された関数が呼出し元のレジスタを破壊しないためにはど
うすればよいのか?

• ra は一つしかないが，関数を多重で呼出す場合は?

• 4つを超える引数の受け渡しは?

これらはスタックと呼ばれるデータ構造をメモリ内に構築することで
取り扱われる



関数呼出時のプログラムの流れ

int g() {

…

}

int f() {

…

g();

}

int main() {

…

f();

…

}

main

f呼出

$s0 を使っており，f の呼出
後も使い続けるとする

こちらでも
$s0 を使うと
どうなる?f の呼出時に

$ra がセット
される

g の呼出時に
$ra は上書きさ
れてしまう

g呼出 リターン

リターン

スタックメモリ

0x8020

0x8010

0x8000

$sp 0x8020

0x8020

0x8010

0x800c

0x8000

$sp 0x800c $sp 0x8020

int f() {

g();

}

f() の

スタック
フレーム

g() の

スタック
フレーム

f() の

スタック
フレーム

0x8020

0x8010

0x8000

f() の

スタック
フレーム

push pop

addu $sp, $sp, -20 addu $sp, $sp, 20



MIPSの場合のレジスタ・スタック利用ルール

• 関数呼出時のレジスタ利用規約
• t0～t9 は，壊されたくないなら呼出し側がスタックに退避
（すなわち，主にテンポラリ用と想定されている）

• s0～s8 は，呼出された側が使いたいならスタックに退避
してリターン時に原状回復する

• 関数を多重で呼出す場合は ra もスタックに退避

• 4つを超える引数はスタックに積んでから関数を呼出す

典型的なメモリマップ

スタックポインタ
($sp)

スタック
（局所変数）

低位アドレス↓

高位アドレス↑

プログラム

大域変数・静的変数等

int global_var;

void func() {

int a, b;

static int static_var; 

...

}

int main() {

int x, y;

int *p = (int *) malloc(256);

}

動的確保領域

グローバルポインタ
($gp)



例: ポケットモンスター 「5かい」バグ

ゲームボーイ用ゲーム「ポケットモンスター 赤」「ポケットモンスター
緑」 （任天堂, 1996） には，プレイヤが保有している道具やポケモ
ン（ゲーム世界内の架空生物）などを並べ替える機能があったが，
この機能の実装に不具合があり，特定の操作によって本来は存在
しないはずの道具（俗に「バグアイテム」）を入手することができた．

道具を使ったときに生じる効果は，道具ごとに定められているアド
レスに対する呼出しによって実現されており，バグアイテムを使用
すると想定外のアドレスへの呼出しが生じた．特に「5かい」という
名称で表示されるバグアイテムの場合は，呼出されるアドレス以
降がポケモンの種類やその状態等を保持するメモリ領域となって
おり，プレイヤがある程度自由に設定することができた．結果とし
て，任意のプログラムを作成し実行することが可能となっていた．

https://www.youtube.com/watch?v=IJ7mRJISeO0

https://www.youtube.com/watch?v=baXxP6b7ANQ

例: バッファ・オーバラン攻撃

int getnum() {

char buf[8];

gets(buf); /* 入力文字列を配列 buf に保存 */

return atoi(buf); /* 文字列 → 整数 */

}

buf

return

address

低位アドレス↓

高位アドレス↑

getNum() の
スタックフレーム

sp

このような問題があるので，gets, 

scanf のように配列長を指定せ
ずに入力を読み込む関数は，使
用が推奨されない

例題(1)

以下に示すプログラムは，ある値を2 進数で表した際に，その中に
含まれる「1」のビットの数を数えるものである．

lw   $s0, 0($s1)

move $t0, $zero

L1: and  $t1, $s0, 1

addu $t0, $t0, $t1

srl  $s0, $s0, 1

bne  $s0, $zero, L1

sw   $t0, 0($s1)

レジスタ $s1 の内容が指すアドレスに値13 が格納されている状
態でこのプログラムを実行した．

(1) 実行終了時の，レジスタ $s0，$t0，$t1 の内容を示せ．
(2) ラベルL1 で指される命令は，何回実行されるか答えよ．

解答例

lw $s0, 0($s1)

move $t0, $zero

L1: 

and  $t1, $s0, 1

addu $t0, $t0, $t1

srl $s0, $s0, 1

bne $s0, $zero, L1

sw $t0, 0($s1)

(1) $s0 は計算対象で，1ビットずつシフトして行き，最後は 0 になる
$t0 は最終結果で，ビット1の数になる．よって 3

$t1 は計算途中で使用し，ループごとに $s0 の最下位ビットを取
り出すのに使われる．最後は 1 になる

(2) 4回

$s0 = mem[$s1 + 0]; /* = 12 */

$t0 = 0;

do {

$t1 = $s0 & 1;

$t0 = $t0 + $t1;

$s0 = $s0 >> 1;

} while ($s0 != 0);

mem[$s1 + 0] = $t0; 

例題(2)

move    $v0, $zero

L1:     lw      $t0, 0($s0)           … (※1)

sltu    $t1, $v0, $t0

beq     $t1, $zero, L2

move    $v0, $t0              … (※2)

L2:     addu    $s0, $s0, 4

addu    $s1, $s1, -1

bne     $s1, $zero, L1

以下に示すプログラムは配列の中から最大値を探すものである．

いま，符号なし整数 (4バイト) が，メモリ上のアドレスが増える方向に
10, 20, 3, 22, 5

の順に並んでおり，この配列の先頭アドレスを $s0, 配列長 5 を $s1 に与えて
このプログラムを実行した．
(1) プログラムの実行が終わった時点でのレジスタ $s1, $v0, $t0, $t1 の内容を

示せ．
(2) ※1 及び※2 の命令がそれぞれ何回実行されるか答えよ．

解答例

move $v0, $zero

L1: lw $t0, 0($s0)

sltu $t1, $v0, $t0

beq $t1, $zero, L2

move $v0, $t0

L2: addu  $s0, $s0, 4

addu  $s1, $s1, -1

bne $s1, $zero, L1

$v0 = 0;

do {

$t0 = mem[$s0 + 0];

if ($v0 < $t0) {

$v0 = $t0;

}

$s0 = $s0 + 4;

$s1 = $s1 – 1;

} while ($s1 != 0)

(1) $s1は処理すべき配列要素の残り数で，終了時は0になる．$v0 はその時
点までの最大値を保持するレジスタで，終了時は全体の最大値22になる．
$t0 は読み出した配列要素を格納するレジスタで，終了時は最後の要素 5 

になる．$t1 は，それまでの最大値が読み出した配列要素より小さければ
1になるレジスタで(これが0のときはL2に分岐するため，最大値の更新が
スキップされる)，最後のループでは 0 になる．

(2) ※1 は，配列長が5なので5回実行される．※2 は，最大値を更新するとき
のみ実行されるため3回実行される．

例題(3)

move    $s2, $zero

L1:     lw      $t0, 0($s0)

addu    $s1, $s1, -1

addu    $s2, $s2, $t0

addu    $s0, $s0, 4

bne     $s1, $zero, L1

以下に示すプログラムは配列の内容の総和をレジスタ $s2 に返すものである．

いま，整数 (4バイト) の配列の先頭アドレスをレジスタ $s0 に，配列長 1000 を
レジスタ $s1 に与え，このプログラムを実行した．
(1) プログラムの実行が終わった時点でのレジスタ $s1 の内容を示せ．
(2) ラベル L1 で指されるlw命令は，何回実行されるか答えよ．
(3) このプログラム全体で，いくつの命令が実行されるか答えよ．(同じ命令が

2 回実行される場合，2 命令と数える)

解答例

(1) $s1は処理すべき配列要素の残り数で，終了時は0になる．
(2) 配列要素を読み出す命令で，配列長分の1000回実行される．
(3) 最初の move 命令は1回だけ，他の5命令は1000回実行され

るので，合計5001命令が実行される．


